
1 Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 17.1 which is part of the IMS test suite in the ‘iwd-TTCN3-B2015-03_D16wk24’ ATS delivery.

The test case can be demonstrated to run one IMS UE (see below). Execution log is provided as evidence.

2 Verification Test Summary

Test Case:
TC_17_1
ATS Version:
iwd-TTCN3-B2015-03_D16wk24
System Simulator used:
R&S CMW500
UE used:
Mediatek Jade MTK6755
Verification Status:
PASS

3 Corrections required for test case 17.1
Change 1 
	Function name
	f_IMS_MOCallSetup_Video_Step3_4

	Reason for change
	183 Session Progress response sent in step 4 needs to carry the precondition option tag in the Require (rather than in the Supported) header. 

	Summary of change
	Added tag to the call of f_IMS_InviteResponse_183_MessageHeaderTX which then puts it into the Require header. Removed it from the Supported header.

	TTCN module
	Common/IMS/IMS_Procedures_Video.ttcn 

	MCC160 Comment
	


Before change

  function f_IMS_MOCallSetup_Video_Step3_4(VideoCallTypeMO_Type p_VideoCallType,

                                           IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                           charstring p_ContactUri := px_IMS_CalleeContactUri,

                                           template (omit) SemicolonParam_List p_AditionalContactParams := omit,

                                           IMS_CallType_Type p_CallType := NORMAL_CALL) runs on IMS_PTC return MessageHeader

  { /* return MessageHeader of the 183 response */

    /* @sic R5-155179: p_CallType sic@ */

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_MessageStep2 := p_InviteRequestWithSdp.SdpOffer;

    var SDP_media_desc v_SDP_Media_AudioRx := v_SDP_MessageStep2.media_list[0];

    var SDP_media_desc v_SDP_Media_VideoRx := v_SDP_MessageStep2.media_list[1];

    var template (value) SDP_Message v_SDP_MessageStep4 := f_IMS_BuildSDP_MOCallAudioVideo_Step4(p_VideoCallType, v_SDP_Media_AudioRx, v_SDP_Media_VideoRx);

    var template (value) MessageHeader v_MessageHeader_Response183;

    // Step 3. SS sends a 100 Trying provisional response.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    // Step 4. Send 183 Session In Progress.

    select (p_VideoCallType) {

      case (C25) {

        v_MessageHeader_Response183 := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequest, tsc_OptionTagList_precondition, p_ContactUri, p_AditionalContactParams, p_CallType); /* @sic R5-142951 sic@ */

      }

      case (AddVideo) {

        v_MessageHeader_Response183 := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequest, -, p_ContactUri, p_AditionalContactParams, p_CallType);

        v_MessageHeader_Response183.supported := cs_Supported(tsc_OptionTagList_precondition);

      }

    }

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine183, v_MessageHeader_Response183, cs_MessageBody_SDP(v_SDP_MessageStep4))));

    return valueof(v_MessageHeader_Response183);

  }
After change

  function f_IMS_MOCallSetup_Video_Step3_4(VideoCallTypeMO_Type p_VideoCallType,

                                           IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                           charstring p_ContactUri := px_IMS_CalleeContactUri,

                                           template (omit) SemicolonParam_List p_AditionalContactParams := omit,

                                           IMS_CallType_Type p_CallType := NORMAL_CALL) runs on IMS_PTC return MessageHeader

  { /* return MessageHeader of the 183 response */

    /* @sic R5-155179: p_CallType sic@ */

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_MessageStep2 := p_InviteRequestWithSdp.SdpOffer;

    var SDP_media_desc v_SDP_Media_AudioRx := v_SDP_MessageStep2.media_list[0];

    var SDP_media_desc v_SDP_Media_VideoRx := v_SDP_MessageStep2.media_list[1];

    var template (value) SDP_Message v_SDP_MessageStep4 := f_IMS_BuildSDP_MOCallAudioVideo_Step4(p_VideoCallType, v_SDP_Media_AudioRx, v_SDP_Media_VideoRx);

    var template (value) MessageHeader v_MessageHeader_Response183;

    // Step 3. SS sends a 100 Trying provisional response.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    // Step 4. Send 183 Session In Progress.

    select (p_VideoCallType) {

      case (C25) {

        v_MessageHeader_Response183 := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequest, tsc_OptionTagList_precondition, p_ContactUri, p_AditionalContactParams, p_CallType); /* @sic R5-142951 sic@ */

      }

      case (AddVideo) {

        v_MessageHeader_Response183 := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequest, tsc_OptionTagList_precondition, p_ContactUri, p_AditionalContactParams, p_CallType);

        //REMOVED v_MessageHeader_Response183.supported := cs_Supported(tsc_OptionTagList_precondition);
      }

    }

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine183, v_MessageHeader_Response183, cs_MessageBody_SDP(v_SDP_MessageStep4))));

    return valueof(v_MessageHeader_Response183);

  }
Change 2 
	Function name
	f_IMS_BuildSDP_MOCallAudioVideo_Step5or7

	Reason for change
	1. Subsequent requests carrying an SDP offer with precondition lines need to allow both “optional” and “mandatory” as strength tag for the remote leg (this will be supported by a prose CR)
2. Usage of the variables to indicate the precondition attributes for audio vs video is the wrong way around.

	Summary of change
	1. Allowed the strength of the remote end to be either optional or mandatory
2. Swapped precondition variables

	TTCN module
	Common/IMS/IMS_SDP_MessagesVideo.ttcn 

	MCC160 Comment
	


Before change

  function f_IMS_BuildSDP_MOCallAudioVideo_Step5or7(VideoCallTypeMO_Type p_VideoCallType,

                                                    SDP_Message p_SDP_MessageInvite) runs on IMS_PTC return template (present) SDP_Message

  { /* @sic R5-153610: p_SDP_MessageInvite instead of p_FmtAudio, p_FmtVideo sic@ */

    var SDP_media_desc v_SDP_Media_InviteAudio := p_SDP_MessageInvite.media_list[0];

    var SDP_media_desc v_SDP_Media_InviteVideo := p_SDP_MessageInvite.media_list[1];

    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(v_SDP_Media_InviteAudio);  // @sic R5-153610 sic@

    var charstring v_FmtVideo := f_SDP_MediaDescr_GetH264_90000(v_SDP_Media_InviteVideo);  // @sic R5-153610 sic@

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio := {

      cr_SDP_Attribute_rtpmap(v_FmtAudio, cr_RTPMAP_AMR_8000),     // @sic R5s130681 change 2 sic@

      cr_SDP_Attribute_fmtp(v_FmtAudio)

    };

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Audio;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Video;

    var template (present) charstring v_DirectionRemote_Audio;

    var template (present) charstring v_DirectionRemote_Video;

    var template (present) charstring v_StrengthRemote := c_optional;

    v_SDP_MediaAttributes_Video := {    /* @sic R5s141336 change 3 sic@ */

      cr_SDP_Attribute_rtpmap(v_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(v_FmtVideo, cr_Video_Fmtp_DefaultParametersMO)  /* @sic R5-150734, R5-150736: same parameters for C.25 and 17.1 sic@ */

    };

    select (p_VideoCallType) {

      case (C25) {

        if (f_SDP_AttributeList_CheckAttribute(v_SDP_Media_InviteAudio.attributes, cr_SDP_Attribute_inactive)) {               /* @sic R5-153610 sic@ */

          v_SDP_MediaAttributes_Audio := f_SDP_Attributes_Add_RX(v_SDP_MediaAttributes_Audio, cr_SDP_Attribute_sendrecv);

        }

        if (f_SDP_AttributeList_CheckAttribute(v_SDP_Media_InviteVideo.attributes, cr_SDP_Attribute_inactive)) {               /* @sic R5-153610 sic@ */

          v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_RX(v_SDP_MediaAttributes_Video, cr_SDP_Attribute_sendrecv);

        }

        v_DirectionRemote_Audio := c_none;

        v_DirectionRemote_Video := c_none;

        v_StrengthRemote := (c_optional, c_mandatory);             /* @sic R5s141221 change 4.1; R5s150236 sic@ */

      }

      case (AddVideo) {

        v_DirectionRemote_Audio := c_sendrecv;

        v_DirectionRemote_Video := c_none;

      }

      case (CallHold) {

        v_DirectionRemote_Audio := c_sendrecv;

        v_DirectionRemote_Video := c_sendrecv;

        v_StrengthRemote := (c_optional, c_mandatory);             /* @sic R5s150755 sic@ */

      }

    }

    v_SDP_PrecondionAttributes_Video := cr_SDP_PrecondionAttributes(c_sendrecv, v_DirectionRemote_Audio, c_mandatory, v_StrengthRemote);

    v_SDP_PrecondionAttributes_Audio := cr_SDP_PrecondionAttributes(c_sendrecv, v_DirectionRemote_Video, c_mandatory, v_StrengthRemote);

    return f_IMS_BuildSDP_AudioVideo_RX(-,

                                        cr_SDP_Media_Audio(v_FmtAudio),

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes_Audio,

                                        cr_SDP_Media_Video(v_FmtVideo),

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes_Video);

  }
After change

  function f_IMS_BuildSDP_MOCallAudioVideo_Step5or7(VideoCallTypeMO_Type p_VideoCallType,

                                                    SDP_Message p_SDP_MessageInvite) runs on IMS_PTC return template (present) SDP_Message

  { /* @sic R5-153610: p_SDP_MessageInvite instead of p_FmtAudio, p_FmtVideo sic@ */

    var SDP_media_desc v_SDP_Media_InviteAudio := p_SDP_MessageInvite.media_list[0];

    var SDP_media_desc v_SDP_Media_InviteVideo := p_SDP_MessageInvite.media_list[1];

    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(v_SDP_Media_InviteAudio);  // @sic R5-153610 sic@

    var charstring v_FmtVideo := f_SDP_MediaDescr_GetH264_90000(v_SDP_Media_InviteVideo);  // @sic R5-153610 sic@

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio := {

      cr_SDP_Attribute_rtpmap(v_FmtAudio, cr_RTPMAP_AMR_8000),     // @sic R5s130681 change 2 sic@

      cr_SDP_Attribute_fmtp(v_FmtAudio)

    };

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Audio;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Video;

    var template (present) charstring v_DirectionRemote_Audio;

    var template (present) charstring v_DirectionRemote_Video;

    var template (present) charstring v_StrengthRemote := c_optional;

    v_SDP_MediaAttributes_Video := {    /* @sic R5s141336 change 3 sic@ */

      cr_SDP_Attribute_rtpmap(v_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(v_FmtVideo, cr_Video_Fmtp_DefaultParametersMO)  /* @sic R5-150734, R5-150736: same parameters for C.25 and 17.1 sic@ */

    };

    select (p_VideoCallType) {

      case (C25) {

        if (f_SDP_AttributeList_CheckAttribute(v_SDP_Media_InviteAudio.attributes, cr_SDP_Attribute_inactive)) {               /* @sic R5-153610 sic@ */

          v_SDP_MediaAttributes_Audio := f_SDP_Attributes_Add_RX(v_SDP_MediaAttributes_Audio, cr_SDP_Attribute_sendrecv);

        }

        if (f_SDP_AttributeList_CheckAttribute(v_SDP_Media_InviteVideo.attributes, cr_SDP_Attribute_inactive)) {               /* @sic R5-153610 sic@ */

          v_SDP_MediaAttributes_Video := f_SDP_Attributes_Add_RX(v_SDP_MediaAttributes_Video, cr_SDP_Attribute_sendrecv);

        }

        v_DirectionRemote_Audio := c_none;

        v_DirectionRemote_Video := c_none;

        v_StrengthRemote := (c_optional, c_mandatory);             /* @sic R5s141221 change 4.1; R5s150236 sic@ */

      }

      case (AddVideo) {

        v_DirectionRemote_Audio := c_sendrecv;

        v_DirectionRemote_Video := c_none;

        v_StrengthRemote := (c_optional, c_mandatory);
      }

      case (CallHold) {

        v_DirectionRemote_Audio := c_sendrecv;

        v_DirectionRemote_Video := c_sendrecv;

        v_StrengthRemote := (c_optional, c_mandatory);             /* @sic R5s150755 sic@ */

      }

    }

    v_SDP_PrecondionAttributes_Audio := cr_SDP_PrecondionAttributes(c_sendrecv, v_DirectionRemote_Audio, c_mandatory, v_StrengthRemote);

    v_SDP_PrecondionAttributes_Video := cr_SDP_PrecondionAttributes(c_sendrecv, v_DirectionRemote_Video, c_mandatory, v_StrengthRemote);

    return f_IMS_BuildSDP_AudioVideo_RX(-,

                                        cr_SDP_Media_Audio(v_FmtAudio),

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes_Audio,

                                        cr_SDP_Media_Video(v_FmtVideo),

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes_Video);

  }
Change 3 
	Function name
	f_IMS_MOCall_ReleaseVideo_ReceiveInvite

	Reason for change
	Strength tags need to be modelled just as above: mandatory for the local leg, and optional or mandatory for the remote leg.
Also, codec name in media line needs to be aligned with RTP/AVPF

	Summary of change
	1. Added mandatory to remote leg for audio
2. Replaced optional by mandatory for local leg for video

3. Addedd optional to remote leg for video

4. Replaced usage of template c_rtpAvp by c_rtAvpf

	TTCN module
	IMS/17/IMS_CC_MediaTestcases.ttcn 

	MCC160 Comment
	


Before change

  function f_IMS_MOCall_ReleaseVideo_ReceiveInvite(charstring p_FmtAudio,

                                                   charstring p_FmtVideo) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  {

    var IMS_DATA_REQ v_IMS_INVITE_REQ;

    var SDP_Message v_SDP_Message;

    var boolean v_A11 := pc_IMS_Video_FeatureTag;

    var boolean v_A12 := false;   // as we have video but not voice

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio := {

      cr_SDP_Attribute_rtpmap(p_FmtAudio, cr_RTPMAP_AMR_8000),

      cr_SDP_Attribute_fmtp(p_FmtAudio)

    };

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video := {

      cr_SDP_Attribute_rtpmap(p_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(p_FmtVideo)

    };

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Audio := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_optional);

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Video := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_optional, c_mandatory);

    var template (present) SDP_Message v_SDP_MessageTemplate := f_IMS_BuildSDP_AudioVideo_RX(cr_SDP_Time_Any,

                                                                                             cr_SDP_Media_Audio,

                                                                                             v_SDP_MediaAttributes_Audio,

                                                                                             v_SDP_PrecondionAttributes_Audio,

                                                                                             cr_SDP_Media(c_video, 0, c_rtpAvp, cr_SDP_FmtList_AtLeastOne(p_FmtVideo)),  /* port number == 0 => release video stream */

                                                                                             v_SDP_MediaAttributes_Video,

                                                                                             v_SDP_PrecondionAttributes_Video);

    v_IMS_INVITE_REQ := f_IMS_INVITE_ReceiveRequest(A_2_1_A5, tsc_OptionTagList_precondition, -, -, -, -, v_A11, v_A12);

    v_SDP_Message := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_INVITE_REQ.Request.Invite.messageBody, v_SDP_MessageTemplate);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_INVITE_REQ, v_SDP_Message);

  }
After change

  function f_IMS_MOCall_ReleaseVideo_ReceiveInvite(charstring p_FmtAudio,

                                                   charstring p_FmtVideo) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  {

    var IMS_DATA_REQ v_IMS_INVITE_REQ;

    var SDP_Message v_SDP_Message;

    var boolean v_A11 := pc_IMS_Video_FeatureTag;

    var boolean v_A12 := false;   // as we have video but not voice

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio := {

      cr_SDP_Attribute_rtpmap(p_FmtAudio, cr_RTPMAP_AMR_8000),

      cr_SDP_Attribute_fmtp(p_FmtAudio)

    };

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video := {

      cr_SDP_Attribute_rtpmap(p_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(p_FmtVideo)

    };

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Audio := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, (c_optional, c_mandatory));
    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes_Video := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, (c_optional, c_mandatory) );

    var template (present) SDP_Message v_SDP_MessageTemplate := f_IMS_BuildSDP_AudioVideo_RX(cr_SDP_Time_Any,

                                                                                             cr_SDP_Media_Audio,

                                                                                             v_SDP_MediaAttributes_Audio,

                                                                                             v_SDP_PrecondionAttributes_Audio,

                                                                                             cr_SDP_Media(c_video, 0, c_rtpAvpf, cr_SDP_FmtList_AtLeastOne(p_FmtVideo)),  /* port number == 0 => release video stream */ 

                                                                                             v_SDP_MediaAttributes_Video,

                                                                                             v_SDP_PrecondionAttributes_Video);

    v_IMS_INVITE_REQ := f_IMS_INVITE_ReceiveRequest(A_2_1_A5, tsc_OptionTagList_precondition, -, -, -, -, v_A11, v_A12);

    v_SDP_Message := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_INVITE_REQ.Request.Invite.messageBody, v_SDP_MessageTemplate);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_INVITE_REQ, v_SDP_Message);

  }
4 Execution Log Files
The Mediatek Jade MTK6755 UE passed this test case on R&S CMW500. The documentation below is enclosed as evidence of the successful test case run:

Test Case Execution log file: R5s160605
